
Python4Delphi80
Python4Delphi

Purpose: To become familiar with the use of Python4Delphi

What is Python4Delphi?
Python1 for Delphi (Python4Delphi) is a set of free components that wrap up 
the Python dll so that Python scripts can be executed from Delphi. 
Figure 80.1 shows a simple demonstration in which a Python script 
BubbleSort.py is loaded and executed from a Delphi program  
Demo1Project.exe.  
The upper window shows the version of python being used (Python 3.8), 
then the output, [0, 1, 2, 3, 5, 8, 9], from the execution of the python script, 
BubbleSort.py, shown in the lower window, acting on the unsorted list,  
[5, 1, 2, 3, 9, 8, 0].

1 "Python" is a registered trademark of the Python Software Foundation

Information
The programs in 
this chapter rely on 
Python4Delphi software, 
inspiration being derived 
from the software 
demos provided with 
Python4Delphi.
For licensing see
https://github.
com/pyscripter/
python4delphi/blob/
master/LICENSE

Figure 80.1 Demo1Project.exe executing Python script BubbleSort.py

Upper window - Memo1

Lower window - Memo2

Unsorted list

Python script

Sorted list

Setting up Python4Delphi
Download python4delphi code, python4delphi-master.zip, or clone this code from the GitHub repository at  
github.com/pyscripter/python4delphi. Unzip it to create a 
python4delphi-master folder with the contents shown in 
Figure 80.2 (September 2020).

1. Start RAD Studio.
2. Add the source subdirectory (e.g., C:\Users\drbond\

python4delphi-master\Source) to the IDE's library
path for the targets you are planning to use
(Tools|Options|Language|Delphi|Library) - Figure
80.3.

3. Open the Python4Delphi package specific to the
version of Delphi being used. For Delphi 10.4.1 and
later, use the package in the Packages\Delphi\Delphi
10.4+ directory. For earlier versions including
Delphi 10.4 (version before 10.4.1) use the package
in the Packages\Delphi\Delphi 10.3- directory
- Figure 80.4. This will need editing for Delphi
10.4. Remove all the {$IFDEF ....}{$LIBSUFFIX 
....} {$ENDIF} statements and replace with 
{$LIBSUFFIX ‘270'}. Figure 80.5 shows the 

Projects pane in the Delphi IDE after opening the 
the package for Delphi 10.4.

4. Python was installed as a part of an Anaconda
distribution (32-bit), Anaconda3, but Python can
also be installed directly from Python.org. You must 
also add the \Library\bin path of the Anaconda distribution to the system environment variable - Figure 80.6.

Figure 80.2 python4delphi_master folder contents

Figure 80.3 Setting Library path

Figure 80.4 Open Project Python_D.dprog

Figure 80.5 Projects pane after opening Python_D.dproj

Figure 80.6 System environment variable

Path to \Library\bin

5. Right click on Python_D270.bpl to bring up the menu shown in Figure 80.7. Click on Install to install the 
python components. If successful, the component palette should now contain a Python palette of components 
as shown in Figure 80.8.

First project
1. Create a new Windows VCL Application for 

Delphi.
2. Drop a TPanel component on the form and 

set its Align property to alTop in the Object 
Inspector. Change its Caption property to 
Python Source Code.

3. Drop a TMemo component (Memo1) on the 
form and set its Align property to alTop. Open its Lines 
property and delete Memo1 string to leave the memo box 
empty. Set ScrollBars to ssBoth.

4. Drop a TSplitter component from the Additional palette on the 
form and set its Align property to alTop and its Height property to 2.

5. Drop a TPanel component on the form and set its Align 
property to alTop in the Object Inspector. Change its 
Caption property to Python Output.

6. Drop a TPanel component on the form and set its Align 
property to alBottom in the Object Inspector. Empty its 
Caption property.

7. Drop a TMemo component (Memo2) on the form and set its 
Align property to alClient so that it fills the gap between the bottom panel and the TSplitter component. 
Open its Lines property and delete Memo2 string to leave the memo box empty. Set ScrollBars to ssBoth.

8. Select the bottom panel and then drop a TButton component (Button1) onto this panel. Set its caption to 
Execute. Set its Anchors property to [akTop,akRight, akBottom].

The user interface's appearance in the design window should now be as shown in Figure 80.9.

Figure 80.7 Install components from here

Right mouse click 
Python_D270.Bpl

Click

Part of the 
menu that 
appears

Figure 80.8 Component palette showing 
installed Python components

Figure 80.9 Design of the user interface

TPanel

TPanel

TPanel

TMemo

TButton

TSplitter

9. Drop a TPythonEngine component from the Python palette on the form. This 
component provides the connection to Python or rather the Python API. Its default 
name is PythonEngine1.

10. Drop a TPythonGUIInputOutput component on the form. Its default name 
is PythonGUIInputOutput1. This component provides a conduit for routing 
input and output between the Graphical User Interface (GUI) and the currently 
executing Python script. 

11. As this project uses an Anaconda distribution (see later for the same project implemented with Python 
from www.Python.org) PythonEngine1's properties are required to be set up as follows

• Set property DllName to python38.dll (because Python 3.8 is installed) or the version that you have installed.

• Set property DllPath to the Anaconda distribution, e.g. c:\users\drbond\Anaconda3.

• Set property AutoLoad to False.

• Set UseLastKnownVersion to False.

• Set PythonEngine1's property IO to PythonGUIInputOutput1.

12. Set PythonGUIInputOutput1's property Output to Memo2.

13. As this Delphi application relies on an Anaconda distribution, we are required to set up a Form Create 
event handler with the following two lines of code (this is unnecessary with Python from Python.org)

Procedure TForm1.FormCreate(Sender: TObject);
  Begin
    PythonEngine1.SetPythonHome('c:\drbond\users\Anaconda3');
    PythonEngine1.LoadDll;
  End;

14. Double click Button1 and add the line of code to the 
event handler

PythonEngine1.ExecStrings(Memo1.Lines);

15. Save the project and its unit in folder FirstExample. 
Name project FirstExampleProject and its unit 
FirstExampleUnit.

16. Now compile, link and run (F9) the executable.

17. Write print(2 + 2) in the Python Source Code 
window then click Execute. Figure 80.10 shows the 
result. The executable calls up the Python interpreter 
which executes the print function with the given 
argument 2 + 2 returning the result 4. This result is 
passed to the Delphi executable which then displays it 
in the Python Output window.

The TSplitter component allows the sizes of the two windows 
to be adjusted together so that as one is enlarged the other is 
reduced in size accordingly seamlessly - Figure 80.11.

Figure 80.11

Figure 80.10 FirstExampleProject.exe in execution

In order to be able to load and save Python scripts for execution by FirstExampleProject.exe, two more 
TButton components need to be added to the bottom panel as shown in Figure 80.12.

18. Select the bottom panel then add two TButton 
components to this panel, Button2 and Button3. Set 
the Caption property of Button2 to Save and the 
Caption property of Button3 to Load. Position these 
two buttons as shown in Figure 80.12.

19. Set the Anchors property of Button3 to [akLeft, 
akTop, akBottom].

20. Set the Anchors property of Button2 to [akTop, 
akBottom].

When the application's window is resized, Button3 (Load) will 
remain anchored at the same distance from the left, top and 
bottom edges of the bottom panel. Button 2 (Save) will remain 
anchored at the same distance from the top and bottom edges 
of the bottom panel but it will move to the right or to the left, 
respectively, if the application's window is resized in either of 
these directions. Button1 (Execute) will remain anchored at 
the same distance from the right, top and bottom edges of the 
bottom panel.

21. Add a TOpenDialog and a TSaveDialog to the form.

22. Double click Button2 (Save) and add the following lines of code to the event handler

With SaveDialog1
  Do
    Begin
      If Execute
        Then Memo1.Lines.SaveToFile(FileName);
    End;

23. Double click Button3 (Load) and add the following 
lines of code to the event handler

With OpenDialog1
  Do
    Begin
      If Execute
        Then Memo1.Lines.LoadFromFile(FileName);
    End;

24. Save project and unit in folder FirstExample.

25. Now compile, link and run (F9) the executable.

26. Click Load and select BubbleSort.py.

27. Click Execute.

Figure 80.13 shows the result.

Property of SaveDialog1

Property of OpenDialog1

Figure 80.12 FirstExampleProject with 
two additional buttons Open and Save

Figure 80.13 FirstExampleProject.exe in execution 
and with Python script BubbleSort.py loaded

www.educational-computing.com/DelphiBook/Code/Chapter80/FirstExample.zip

Second project
Save project FirstExampleProject.dproj in a new folder, SecondExample, 
rename the project SecondExampleProject.dproj (Save Project As).
Save unit FirstExampleUnit.pas in new folder SecondExample rename the 
unit SecondExampleUnit.pas (Save As).

1. Select bottom panel and add a TButton component to this panel. Change 
its Caption property to Show Var. Position as shown in Figure 80.14.

2. Add a TPythonDelphiVar component from the Python palette. Change its 
VarName property to test. Set its Engine property to PythonEngine1 if 
this doesn't happen automatically.

3. Double click Show Var and add the following line of code to the event 
handler

ShowMessage('Value = ' + 
PythonDelphiVar1.ValueAsString);

4. Save project and unit.
5. Now compile, link and run (F9) 

the executable.
6. Enter the following Python code 

into the Python Source Code 
window
test.value = 5
print(test, test.value)

as shown in Figure 80.15.
7. Click Execute. The line

<PythonDelphiVar: 5> 5

appears in the Python Output 
window. Python function print 
copies the numeric value 5 
stored in the Python variable 
test.value and then sends 
it to Delphi to handle via 
PythonGUIInputOutput1 
which is also connected to 
Memo2.

8. Now click Show Var to execute 
some Delphi code. The window 
shown in Figure 80.15 should 
appear displaying the string 
'Value = 5'. This is occurs 
because the identifiers 
PythonDelphiVar1 and test 
are aliases for the same variable.

The fact that these identify the same 
variable means it is possible to read and write the variable's content in both Python and Delphi!

Information
The identifiers test and 
PythonDelphiVar1 operate as 
aliases for the same variable. It 
makes better sense therefore to 
use the same identifier name 
for each. PythonDelphiVar1 is 
the property value of property 
Name for a TPythonDelphiVar 
object and the other test is 
the property value of property 
VarName. The latter is the 
variable name that is accessible 
to a Python script.

Figure 80.14 SecondExampleProject user interface design

Figure 80.15 SecondExampleProject.exe in execution

Share common 
value 5

Output from Python script

Output from Delphi 
ShowMessage

www.educational-computing.com/DelphiBook/Code/Chapter80/SecondExample.zip

Third project
Save project SecondExampleProject.dproj in a new folder, ThirdExample, renaming the project 
ThirdExampleProject.dproj (Save Project As).
Save unit SecondExampleUnit.pas in new folder ThirdExample renaming the unit ThirdExampleUnit.pas 
(Save As).

1. Select bottom panel and add a TEdit component, Edit1, to this panel as shown in Figure 80.16. Set its 
Text property to 0.

2. Click on PythonDelphiVar1 and in the Object Inspector switch to the Events tab.
3. Double-click on attribute OnGetData to create its event handler. Add the following line of code to this 

event handler: Data := Edit1.Text;
4. Double-click on attribute OnSetData, and add the following line of code to this event handler:  

               Edit1.Text := Data;

5. Double-click on the OnChange attribute, and add the following line of code to this event handler:
With Sender As TPythonDelphiVar
  Do ShowMessage('Var test changed: ' + PythonDelphiVar.ValueAsString);

6. Save project and unit.
7. Now compile, link and run (F9) the 

executable - Figure 80.17.
8. Write the Python script  

test.value=45 in the Python 
Source Code window.

9. Click Execute. The value shown in 
the text box Edit1 will change from 
its default value 0 to new value 45 
as a consequence of executing the 
given Python script. The outcome 
demonstrates that Python variable 
test is associated with the variable 
Edit1 in the Delphi program.

Figure 80.17 ThirdExampleProject just launchedFigure 80.16 ThirdExampleProject user interface design

Edit1 text box

Figure 80.18 ThirdExampleProject executing script test.value=45

OnChange event handler executed because 
PythonDelphiVar1's value changed

Free sample chapter -  copyright Dr K R Bond 2020

10. Click OK to close the popup window.

11. Delete test.value=45 in the Python Source Code window.

12. Write print(test.value) in the blank Python Source Code window

13. Change the contents of text box Edit1 to 30.

14. Click Execute. The value 30 now appears in the Python Output window reflecting the fact that the 
value assigned to Python variable test has been updated to the value of Delphi text box, Edit1.

This exercise has demonstrated that it is possible using components from the Python palette to create two-way 
communication between a Delphi program in execution and a Python script in execution.

Fourth project
1. Create a new Windows VCL Application for Delphi.

2. Drop a TPanel component on the form and set its Align property to alTop in the Object Inspector. 
Change its Caption property to Python Source Code.

3. Drop a TPanel component on the form and set its Align property to alBottom in the Object Inspector. 
Empty its Caption property.

4. Select the bottom panel and then drop a TButton component (Button1) onto this panel. Set its caption 
to Execute. Set its Anchors property to [akTop,akRight, akBottom].

5. Drop a TMemo component (Memo1) on the form and set its Align property to alClient. Open its Lines 
property and delete Memo1 string to leave the memo box empty. Set ScrollBars to ssBoth.

6. Drop a TPythonEngine component from the Python palette on the form. This component provides the 
connection to Python or rather the Python API. Its default name is PythonEngine1.

7. Drop a TPythonInputOutput component on the form. Its default name is PythonInputOutput1. This 
component provides a conduit for routing input and output between a console window and the currently 
executing Python script.

Figure 80.19 FourthExampleProject GUI design

www.educational-computing.com/DelphiBook/Code/Chapter80/ThirdExample.zip

www.educational-computing.com/DelphiBook/Code/Chapter80/FourthExample.zip

The user interface's appearance in the design window should now be as shown in Figure 80.19. As this project 
uses an Anaconda distribution (see later for the same project implemented with Python from www.Python.org) 
PythonEngine1's properties are required to be set up as follows

• Set property DllName to python38.dll (because Python 3.8 is installed ) or the version that you have 
installed).

• Set property DllPath to the Anaconda distribution, e.g. c:\users\drbond\Anaconda3.

• Set property AutoLoad to False.

• Set UseLastKnownVersion to False.

• Set PythonEngine1's property IO to PythonInputOutput1.

8. Select PythonInputOutput1 and its Events page. Double click OnReceiveData to create event handler 
PythonInputOutputReceiveData.

9. Select PythonInputOutput1 and its Events page. Double click OnSendData to create event handler 
PythonInputOutputSendData.

10. Add the line of code to PythonInputOutputReceiveData 

Readln(Data);

11. Add the line of code to PythonInputOutputSendData 

Writeln(Data);

12. In order to read from and write to the console we need to create a console at run time. Click Project|View 
Source to bring up the application's source code. Add the following directive to the application's source 
code as shown in Figure 80.20

{$APPTYPE CONSOLE}

13. Save the project using filename FourthExampleProject.dproj (Save Project As) in a folder 
FourthExample.

14. Save its unit using filename FourthExampleUnit.pas in the same folder FourthExample.

Program FourthExampleProject;
{$APPTYPE CONSOLE}
Uses
  Vcl.Forms,
  FourthExampleUnit in 'FourthExampleUnit.pas' {Form1};

{$R *.res}

Begin
  Application.Initialize;
  Application.MainFormOnTaskbar := True;
  Application.CreateForm(TForm1, Form1);
  Application.Run;
End.

Figure 80.20 Program FourthExampleProject

Creates Console window

Creates GUI window

Free sample chapter -  copyright Dr K R Bond 2020

15. Now compile, link and run (F9) the executable. Two windows should appear: a console window and an 
application window as shown in Figure 80.21.

16. Enter the following Python script in the Python Source Code window and click Execute
value = input("Please enter a string\n")
print(f'You entered {value}')

17. The prompt Please enter a string should appear in the console window as shown in Figure 80.21.
18. Click the console window and enter Hello World! then press return.  

The string You entered Hello world! shown in Figure 80.21 should be echoed to the console 
window.

Fifth project - this project was inspired by Kiriakos Vlahos (aka PyScripter)
Open project FirstExampleProject.dproj and save in a new folder, 
FifthExample, whilst renaming it FifthExampleProject.dproj (Save Project As).
Save unit FirstExampleUnit.pas 
in folder FifthExample renaming it 
FifthExampleUnit.pas (Save As).

1. Add a TPythonModule to the 
form (the form should already 
contain a TPythonEngine and a 
TPythonGUIInputOutput component).

2. Add the function IsPrime shown in 
Figure 80.22 to the implementation 
section of FifthExampleUnit.pas.

3. Select PythonModule1 and in the Object Inspector click the ellipsis (Figure 80.23) to bring up the Events 
editor shown in Figure 80.24.

Figure 80.21 FourthExampleProject in execution

Function IsPrime(No : Integer) : Boolean;
  Begin
    If (No <= 1) Then Exit(False);
    Var UpperLimit := Floor(Sqrt(No));
    For Var i := 2 To UpperLimit
      Do If (No Mod i = 0) Then Exit(False);
    Exit(True);
  End;

Figure 80.22 Function IsPrime

Figure 80.23 Events property PythonModule1

Figure 80.24 
delphi_is_prime 

event
Figure 80.25 delphi_is_prime 

event

Every composite positive 
integer has a factor other 
than 1 or itself less than or 
equal to the greatest integer 
less than or equal to its 
square root

www.educational-computing.com/DelphiBook/Code/Chapter80/FifthExample.zip

4. Click           in the Events editor to add a new event named delphi_
is_prime. Figure 80.24 already shows that a new event hasalready been 
added and then named delphi_is_prime. This naming is done by 
highlighting the default name of this new event in the Events editor and 
then in the Object Inspector, changing the events Name property to delphi_is_prime - Figure 80.25.

5. In the Structure pane select Events and 0 - delphi_is_prime - Figure 80.26.
6. Switch to the Events page for PythonModule1.

Events[0] and double click the field for the 
OnExecute event to create the event handler 
PythonModule1Events0Execute - Figure 80.27.

7. Switch to the source code view of 
FifthExampleUnit.pas and add the lines of 
source code shown in Figure 80.28 to the event handler PythonModule1Events0Execute.

8. Select PythonModule1 in the Object Inspector and set its ModuleName property to delphi_module.
9. Save all (Shift+Ctrl+S).
10. Now compile, link and run (Shift+Ctrl+F9) the executable.

The following exercise was inspired by a Webinair given by Kiriakos Vlahos (aka PyScripter):

11. Click Load and open countofprimesinpythononly.py (Figure 80.29).
12. Click Execute to run countofprimesinpythononly.py. Repeat this two more times.
13. The Python program counts the number of primes between 0 and 1000000, measures the time that elapses 

and then prints both results - Figure 80.30. The elapsed time is approximately 6.5 seconds.
14. Click Load and open pythondelphiprime.py - see Figure 80.31. This script uses a function, delphi_

is_prime, linked to PythonModule1 via module delphi_module, to test for primality in place of the 
Python is_prime function in countofprimesinpythononly.py. Figure 80.28 shows that delphi_
is_prime in turn relies on function IsPrime written in Delphi. When the python script is run it invokes 
the event handler PythonModule1Events0Execute which in turn calls the Delphi function IsPrime. 

Procedure TForm1.PythonModule1Events0Execute(Sender: TObject; PSelf,
                                             Args: PPyObject; 
                                             Var Result: PPyObject);
  Var
    N : Integer;
  Begin
    With GetPythonEngine
      Do
        Begin
          If PyArg_ParseTuple(Args, 'i:delphi_is_prime', @N) <> 0
            Then
              Begin
                If IsPrime(N)
                  Then Result := PPyObject(Py_True)
                  Else Result := PPyObject(Py_False);
                Py_INCREF(Result);
              End
            Else Result := Nil;
        End;
  End; Figure 80.28 Event handler PythonModule1Events0Execute 

Figure 80.27 Creating OnExecute event handler

Figure 80.26

Free sample chapter -  copyright Dr K R Bond 2020

PythonEngine1 and PythonModule1 
provide the "wiring" between the Python 
script and the Delphi event handler 
PythonModule1Events0Execute. The 
remainder of the code of the latter packs and 
unpacks the communication between Python 
script and the Delphi program.

15. Click Execute to run pythondelphiprime.py. 
Repeat this two more times.

16. The Python program counts the number 
of primes between 0 and 1000000, 
measures the time that elapses and then 
prints both results - Figure 80.32. The 
elapsed time is approximately 0.32 
seconds. Twenty times faster than script 
countofprimesinpythononly.py.

from timeit import Timer

import math

def is_prime(n):

    if n <= 1:

        return False

    upperlimit = math.floor(math.sqrt(n))

    for i in range(2, upperlimit + 1):

        if (n % i== 0):

            return False

    return True

def count_primes(max_n):

    result = 0

    for i in range(2, max_n + 1):

        if is_prime(i):

            result += 1

    return result

def test():

    max_n= 1000000

    print(f'Numberof primes between 0 and {max_n} = {count_primes(max_n)}')

def main():

    print(f'Elapsedtime: {Timer(stmt=test).timeit(1)} secs')

if __name__ == '__main__':

    main() Figure 80.29 countofprimesinpythononly.py 

Figure 80.30 countofprimesinpythononly.py executed three times

Free sample chapter -  copyright Dr K R Bond 2020

Sixth project
This project was inspired by Kiriakos Vlahos  
(aka PyScripter):
1. Save project FifthExampleProject.dproj 

in a new folder, SixthExample, renaming it 
SixthExampleProject.dproj (Save Project 
As). Save unit FirstExampleUnit.pas in folder 
SixthExample renaming it  
SixthExampleUnit.pas (Save As).

2. Select PythonModule1 and in the Object 
Inspector click the ellipsis (Figure 80.23) to bring 
up the Events editor.

3. Click           in the Events editor to add a new 
event delphi_count_primes - Figure 80.33. 

4. In the Structure pane select Events and  
1 - delphi_count_primes.

5. Switch to the Events page for PythonModule1.
Events[1] and double click the field for the 
OnExecute event to create the event handler 
PythonModule1Events1Execute.

6. Switch to the source code view of 
SixthExampleUnit.pas and add the function 
CountPrimes - Figure 80.34.

7. Add the lines of source code shown in Figure 80.35 to the event handler 
PythonModule1Events1Execute.

8. Save all (Shift+Ctrl+S).
9. Now compile, link and run (Shift+Ctrl+F9) the executable.
10. Click Load and open CountoPrimesParallel.py (Figure 80.36).

from delphi_module import delphi_is_prime

from timeit import Timer

import math

def count_primes(max_n):

    result=0

    for i in range(2,max_n+1):

        if delphi_is_prime(i):

            result +=1     

    return result

def test():

    max_n=1000000

    print(f'Number of primes between 0 and {max_n}={count_primes(max_n)}')

def main():

    print(f'Elapsed time:{Timer(stmt=test).timeit(1)}secs')

if __name__=='__main__':

    main() Figure 80.31 pythondelphiprime.py 

Figure 80.32 pythondelphiprime.py executed three times

Figure 80.33 Events editor, new 
event delphi_count_primes

www.educational-computing.com/DelphiBook/Code/Chapter80/SixthExample.zip

11. Click Execute to run CountoPrimesParallel.py. Repeat this two more times.
12. The Python program counts the number of primes between 0 and 1000000, measures the time that elapses 

and then prints both results - Figure 80.37. The elapsed time is approximately 0.06 seconds which is a 100 
times faster than the python script countofprimesinpythononly.py.

TParallel provides a class for-loop, &For, which makes efficient use of all the CPU cores in the system.

Function CountPrimes(MaxN : Integer): Integer;
  Begin
    Var Count := 0;
    TParallel.&For(2, MaxN, Procedure(i : Integer)
                              Begin
                                If IsPrime(i)
                                  Then AtomicIncrement(Count);
                               End
                   );
    Result := Count;
  End;

Figure 80.34 Function CountPrimes

Procedure TForm1.PythonModule1Events1Execute(Sender: TObject; PSelf,
                                             Args: PPyObject; 
                                             Var Result: PPyObject);
  Var
    N : Integer;
  Begin
    With GetPythonEngine
      Do
        Begin
          If PyArg_ParseTuple(Args, 'i:delphi_count_primes', @N) <> 0
            Then
              Begin
                Result := PyLong_FromLong(CountPrimes(N));
                Py_INCREF(Result);
              End
            Else Result := Nil;
        End;
  End; Figure 80.35 Event handler PythonModule1Events1Execute

from delphi_module import delphi_count_primes 
from timeit import Timer
import math
def test():
    max_n = 1000000
    print(f'Number of primes between 0 and {max_n} = {delphi_count_primes(max_n)}')
def main():
    print(f'Elapsed time:{Timer(stmt=test).timeit(1)}secs')
if __name__=='__main__':
    main() Figure 80.36 CountPrimesParallel.py

Free sample chapter -  copyright Dr K R Bond 2020

Using Python4Delphi with a version of Python installed from Python.org
Download Windows x86-64 executable installer for Python from 
python.org/downloads/windows - Figure 80.38.

Click on python-3.9.0-amd64.exe (or the version that you 
downloaded) to launch the install window. Tick Add Python 3.9 to 
PATH then click Install Now to install Python 3.9 - Figure 80.39.

Python4Delphi should find Python39.dll automatically.

Create a new VCL application and add a TPythonEngine component 
to the form. Figure 80.40 shows the Object Inspector pane for 
PythonEngine1. For this version of Python4Delphi the default 
settings of interest for DelphiPython1 are

• UseLastKnownVersion = True

• DllName = python39.dll

• AutoLoad = True.

When earlier in this chapter we worked with an Anaconda installation 
of Python - Python 3.8 we needed the application to initialise 
PythonEngine1 as follows

  PythonEngine1.SetPythonHome('c:\Users\drbond\Anaconda3');

  PythonEngine1.LoadDll;

Using AutoLoad set to True takes care of this automatically.

Figure 80.37 CountPrimesParallel.py

Figure 80.38 Python 3.9 download versions

Figure 80.39 Python 3.9.0(64-bit) install

Figure 80.40 Objector Inspector

When Python is installed in Windows, the user has the option to register it, either for all users or just for this 
user. Registration involves writing information to the registry about the location of the installation, the name and 
location of the help file etc.

Setting UseLastKnownVersion = True forces 
Delphi to use the latest registered version of 
Python.

If the programmer requires a different 
registered2 version then they need to set the 
following properties:

• DLLName e.g. python38.dll

• RegVersion e.g 3.8

• Set UseLastKnownVersion property to False.

A little more work needs to be done to use a specific unregistered version:

Set
• DLLName e.g. python38.dll

• RegVersion e.g 3.8

• Set UseLastKnownVersion property to False

• Set DLLPath to the path where the DLL is located, e.g. C:\Users\drbond\anaconda3

• Set AutoLoad property to False

• The event handler for the OnCreate event of Form1 (or whatever the form launched by the application 
is called), or in another suitable place, must contain the following code (assuming that PythonEngine1 is 
the name of the component)

PythonEngine1.SetPythonHome('Python installation directory');

PythonEngine1.LoadDLL;

32-bit Delphi applications only work with 32-versions of Python and 64-bit Delphi applications only work with 
64-bit versions of Python.

The project FirstExampleProject created using the Anaconda installed Python may now be recreated Python 3.9 
downloaded directly from www.python.org.

2 Anaconda distributions require that SetPythonHome is called as shown above even if they are registered.

Figure 80.41 Registry entry for 
Computer\HKEY_CURRENT_USER\SOFTWARE\Python\PythonCore

Seventh project
This project manipulates images in Python which are loaded, displayed and saved using Delphi. 

The Python side of things requires a module called pillow (PIL). The easiest way to install pillow is automatically as 
an integral part of an Anaconda installation, installing it directly is quite tricky because it depends on other modules 
being pre- installed. Anaconda automatically takes care of everything. For this reason, the Anaconda installation of 
Python - Python 3.8 - is used in this project. It will therefore be necessary to set up the Python Engine as per the 
instructions for unregistered versions of Python - see previous page.

1. Create a new Windows VCL Application for Delphi 
and save in a new folder SeventhExample with 
filenames SeventhExampleProject.dproj and 
SeventhExampleUnit.pas.

2. Drop a TPanel component on the form and set its Align 
property to alBottom in the Object Inspector. Set its 
Height property to 41 and clear its Caption property.

3. With this panel selected drop three buttons on the panel 
and set the Top property of each to 8. Position the buttons 
as shown in Figure 80.42.

4. Label the buttons Load Image, Save Image and Execute 
according to Figure 80.42 adjusting the width of each 
accordingly.

5. Set the Anchors property of Load Image to [akLeft, 
akTop, akBottom].

6. Set the Anchors property of Save Image to [akTop, 
akBottom].

7. Set the Anchors property of Execute to [akTop, akRight, 
akBottom].

8. Drop a TMemo component (change Name to Memo2) on the form and set its Align property to alBottom. 
Open its Lines property and delete Memo2 string to leave the memo box empty for the moment. Set 
ScrollBars to ssBoth.

9. Drop a TSplitter component from the Additional palette on the form and set its Align property to alBottom and its 
Height property to 2.

10. Drop a TPanel component on the form and set its Align property to alBottom in the Object Inspector. 
Change its Caption property to Python Script.

11. Drop a TMemo component (Memo1) on the form and set its Align property to alBottom. Open its Lines 
property and delete Memo1 string to leave the memo box empty. Set ScrollBars to ssBoth.

12. Drop a TPanel component on the form and set its Align property to alBottom in the Object Inspector. Set 
its Caption property to Image information (Don't exceed 2.5 MB).

13. Drop a TImage component set its Align property to alClient and its Proportional property to True.

14. Drop a TPythonEngine and a TPythonGUIINputOutput component on the form.

15. As the author's Python installation was done via Anaconda, PythonEngine1 needs to be set up as follows

Figure 80.42 SeventhExampleProject in execution

www.educational-computing.com/DelphiBook/Code/Chapter80/SeventhExample.zip

• DLLName e.g. python38.dll
• RegVersion e.g 3.8
• Set UseLastKnownVersion property to False
• Set DLLPath to the path where the DLL is located, e.g. C:\Users\drbond\anaconda3
• Set AutoLoad property to False
• The event handler for the OnCreate event of Form1 (or whatever the form launched by the application 

is called), or in another suitable place, must contain the following code (assuming that PythonEngine1 is 
the name of the component)

PythonEngine1.SetPythonHome('Python installation directory');

PythonEngine1.LoadDLL;

16. Set PythonEngine1's IO property to PythonGUIInputOutput1.
17. Set PythonGUIInputOutput1's Output property to Memo1.
18. Add a TOpenDialog and a TSaveDialog component to the form.
19. Double click Save Image button and add the following lines of code to the event handler

With SaveDialog1
  Do
    Begin
      If Execute
        Then Memo1.Lines.SaveToFile(FileName);
    End;

20. Double click Load Image button and add the following lines of code to the event handler
With OpenDialog1
  Do
    Begin
      If Execute
        Then Memo1.Lines.LoadFromFile(FileName);
    End;

21. Save project and unit in folder SeventhExample.
22. Select Memo2 and open its Lines property so that the 

contents of script ImageProcessingScript.py shown in 
Figure 80.45 can be pasted into the String List Editor.

23. Add a Uses clause to the Implementation section  
(place below {R *.dfm}) as follows

Uses
  VarPyth,
  Math,
  jpeg;

24. Add the source code shown in Figure 80.46 to the 
implementation section.

25. Save all (Shift+Ctrl+S).
26. Now compile, link and run (Shift+Ctrl+F9) the executable.
27. The form shown in Figure 80.43 should show.
28. Use the splitter bar to enlarge the Python script window.
29. The default setting of Python script is new_im = im.convert('L'). This Python statement when 

executed converts an RGB image to a black and white image one.

Property of SaveDialog1

Property of OpenDialog1

Figure 80.43 SeventhExampleProject 
in execution

30. Click Load Image and select from the images folder 
BorneoOrangUtan1SmallRes1.jpg and open - Figure 
80.42.

31.  Click Execute to convert the image to a black and white one 
- Figure 80.44.

32. Comment out new_im = im.convert('L') in the 
Python Script window and uncomment  
         new_im = im.rotate(90, expand=True).

33. Click Load Image and select from the images folder 
FarmPortDicksonLowResSmall.jpg and open.

34. Click Execute to rotate this image through 90° counterclockwise. 
Figure 80.47 shows the rotated image.

35. Click Load Image and select from the images folder 
CockatooSmallLowRes.jpg and open.

36. Comment out  new_im = im.rotate(90, expand=True) in the Python Script window and 
uncomment  new_im = im.filter(FIND_EDGES).

37. Click Execute to find the images edges - Figure 80.48.

#  This demo requires  the module pillow (PIL)
#  The easiest way to install module pillow is to install Anaconda
from io import BytesIO
from PIL import Image, ImageFilter, ImageDraw, ImageFont
from PIL.ImageFilter import( BLUR, CONTOUR, DETAIL, EDGE_ENHANCE, 
                             EDGE_ENHANCE_MORE, EMBOSS,
                             FIND_EDGES,SMOOTH, SMOOTH_MORE, SHARPEN)
import sys

def ProcessImage(data):
    print(sys.version)
    stream = BytesIO(data)
    im = Image.open(stream)
    print ("Processing image %s of %d bytes" % (im.format, len(data)))
    # new_im = im.rotate(90, expand=True)
    # new_im = im.filter(ImageFilter.BLUR)
    # new_im = im.filter(ImageFilter.BoxBlur(5))
    # new_im = im.filter(ImageFilter.GaussianBlur(5))
    # new_im = im.crop((100,200,400,500))
    # new_im = im.transpose(Image.FLIP_LEFT_RIGHT)
    # new_im = im.transpose(Image.FLIP_TOP_BOTTOM)
    # new_im = im.transpose(Image.ROTATE_90)
    # new_im = im.resize(round(im.size[0]*0.5), round(im.size[1]*0.5))
    # width, height = im.size
    # draw = ImageDraw.Draw(im)
    # text = "sample watermark"
    # font = ImageFont.truetype('arial.ttf', 36)
    # textwidth, textheight = draw.textsize(text, font)
    # calculate the x,y coordinates of the text
    # margin = 10
    # x = width - textwidth - margin
    # y = height - textheight - margin
    # draw watermark in the bottom right corner
    # draw.text((x, y), text, font=font)
    new_im = im.convert('L')
    # new_im = im.filter(CONTOUR)
    # new_im = im.filter(EMBOSS)
    # new_im = im.filter(FIND_EDGES)
    new_im.format = im.format
    return new_im  
def ImageToString(image):
    stream = BytesIO()
    image.save(stream, image.format)
    return stream.getvalue()

Figure 80.45 ImageProcessingScript.py

Figure 80.44 Black & white conversion

Function ImageToPyStr(AGraphic : TGraphic) : Variant;
  Var
    _Stream : TMemoryStream;
    _Str : PPyObject;
  Begin
    _Stream := TMemoryStream.Create();
    Try
       AGraphic.SaveToStream(_Stream);
      _Str := GetPythonEngine.PyString_FromStringAndSize(_Stream.Memory, _Stream.Size);
      Result := VarPythonCreate(_Str);
      GetPythonEngine.Py_DECREF(_Str);
    Finally
      _Stream.Free;
    End;
  End;
Procedure TForm1.Button1Click(Sender: TObject);
  Var
    _im : Variant;
    _Stream : TMemoryStream;
    _dib : Variant;
    pargs: PPyObject;
    presult :PPyObject;
    P : PAnsiChar;
    Len : NativeInt;
  Begin
    If (Image1.Picture.Graphic = nil) or Image1.Picture.Graphic.Empty
      Then Raise Exception.Create('You must first select an image');
    PythonEngine1.ExecStrings(Memo2.Lines);
    _im := MainModule.ProcessImage(ImageToPyStr(Image1.Picture.Graphic));
    If Not chkUseDC.Checked
      Then
        Begin
         // We have to call PyString_AsStringAndSize because the image may contain zeros
          With GetPythonEngine
            Do
              Begin
                pargs := MakePyTuple([ExtractPythonObjectFrom(_im)]);
                Try
                  Try
                    presult := PyEval_CallObjectWithKeywords(ExtractPythonObjectFrom(MainModule.ImageToString),
                                                             pargs, nil);
                    If (PyString_AsStringAndSize(presult, P, Len) < 0) or (P = nil)
                      Then
                        Begin
                          ShowMessage('This does not work and needs fixing');
                          Abort;
                        End;
                  Finally
                    Py_XDECREF(pResult);
                  End;
                Finally
                  Py_DECREF(pargs);
                End;
              End;
          _Stream := TMemoryStream.Create();
          Try
            _Stream.Write(P^, Len);
            _Stream.Position := 0;
            Image1.Picture.Graphic.LoadFromStream(_stream);
          Finally
            _Stream.Free;
          End;
        End
      Else
        Begin
          Image1.Picture.Bitmap.SetSize(Image1.Width, Image1.Height);
          _dib := Import('PIL.ImageWin').Dib(_im);
          Image1.Picture.Bitmap.SetSize(Image1.Height, Image1.Width);
          _dib.expose(NativeInt(Image1.Picture.Bitmap.Canvas.Handle));
        End;

Figure 80.46 Source code from Python4Delphi Demo 29

Figure 80.47 Rotated image Figure 80.48 Finding edges

Before

After

TO BE CONTINUED.....

Programming Task
Create a new VCL application which implements image processing based on the Python script shown in Figure 
80.45. 
The user interface should enable a user to
• load an image
• save an image
• select and apply a particular image processing effect, e.g. convert a coloured image to monochrome (black 

and white). The application should then automatically execute the corresponding Python code to achieve this 
effect. No visible reference to this Python code should appear in the user interface when operating normally. 

The application should
• check that an image is loaded before attempting to apply any image processing
• prevent images from loading that are above 1MB in size.

1

